
Le grand livre de eFORTH Linux

version 1.2 - 1 décembre 2023



Auteur

Marc PETREMANN

Collaborateur(s)

• XXX

Table des matières

Installer eForth sous Linux	4
Prérequis	4
Installer eForth Linux sous Linux	
Lancer eForth Linux	5
Un vrai FORTH 64 bits avec eForth Linux	7
Les valeurs sur la pile de données	
Les valeurs en mémoire	
Traitement par mots selon taille ou type des données	
Conclusion	
Edition et gestion fichiers sources pour eForth Linux Les éditeurs de fichiers texte	
Stockage sur GitHub	
Editer des fichiers pour eForth Linux depuis Windows	
Création et gestion de projets FORTH avec Netbeans	
Créer un projet eForth avec Netbeans	
Quelques bonnes pratiques	
Exécution du contenu d'un fichier par eForth Linux	
·	
Le système de fichiers Linux Manipulation des fichiers	
Organiser et compiler ses fichiers avec eForth Linux	
Organiser et compiler ses richiers avec erorut Linux Organiser ses fichiers	
Enchaînement des fichiers	
Conclusion	
Commentaires et mise au point	
Ecrire un code FORTH lisible	
Indentation du code source	
Les commentaires de pile	
Signification des paramètres de pile en commentaires	
Commentaires des mots de définition de mots	
Les commentaires textuels	
Commentaire en début de code source	
Outils de diagnostic et mise au point	
Le décompilateur	
Dump mémoire	
Moniteur de pile	
·	
Dictionnaire / Pile / Variables / Constantes Étendre le dictionnaire	
Gestion du dictionnaire	
Piles et notation polonaise inversée	
Manipulation de la pile de paramètres	
La pile de retour et ses utilisations	

Utilisation de la mémoire	33
Variables	
Constantes	33
Valeurs pseudo-constantes	
Outils de base pour l'allocation de mémoire	34
Les variables locales avec eForth Linux	36
Introduction	
Le faux commentaire de pile	
Action sur les variables locales	
Structures de données pour eForth Linux	40
Préambule	
Les tableaux en FORTH	
Tableau de données 32 bits à une dimension	
Mots de définition de tableaux	
Gestion de structures complexes	41
Les nombres réels avec eForth Linux	
Les réels avec eForth Linux	
Precision des nombres réels avec eForth Linux	
Constantes et variables réelles	
Opérateurs arithmétiques sur les réels	
Opérateurs mathématiques sur les réels	
Opérateurs logiques sur les réels	
Transformations entiers ↔ réels	46
Affichage des nombres et chaînes de caractères	48
Changement de base numérique	
Définition de nouveaux formats d'affichage	
Affichage des caractères et chaînes de caractères	51
Variables chaînes de caractères	
Code des mots de gestion de variables texte	53
Ajout de caractère à une variable alphanumérique	55
Les mots de création de mots	57
Utilisation de does>	
Exemple de gestion de couleur	
Exemple, écrire en pinyin	
Traitement des caractères UTF8	
Le codage UTF8	
Récupérer le code de caractères UTF8 entrés au clavier	
Affichage de caractères UTF8 depuis leur code	

Installer eForth sous Linux

eForth Linux est une version très puissante destinée au système Linux. eForth Linux fonctionne sur toutes les versions récentes de Linux, y compris dans un environnement virtuel Linux.

Prérequis

Vous devez disposer d'un système Linux opérationnel :

- installé sur un ordinateur utilisant Linux comme seul système d'exploitation;
- installé dans un environnement virtuel.

Si vous disposez seulement d'un ordinateur sous Windows 10 ou 11, vous pouvez installer Linux dans le sous-système **WSL**¹.

Le Sous-système Windows pour Linux permet aux développeurs d'exécuter un environnement GNU/Linux (et notamment la plupart des utilitaires, applications et outils en ligne de commande) directement sur Windows, sans modification et tout en évitant la surcharge d'une machine virtuelle traditionnelle ou d'une configuration à double démarrage.

L'intérêt d'une installation d'une distribution Linux dans **WSL** permet d'avoir à disposition une version Linux en mode commande en quelques secondes. Ici, **Ubuntu** est accessible depuis le système de fichiers Windows et se lance en un seul clic :

¹ WSL = Windows Subsystem Linux




Figure 1: Ubuntu accessible en un clic depuis WSL sous Windows

Toutes les instructions pour installer **WSL2** puis la distribution Linux de votre choix sont disponibles ici :

https://learn.microsoft.com/fr-fr/windows/wsl/install

Par défaut, WSL2 propose d'installer la distribution Linux Ubuntu.

Installer eForth Linux sous Linux

Si vous lancez Ubuntu (ou toute autre version de Linux), vous vous retrouverez par défaut dans votre répertoire utilisateur. On commence par accéder au dossier **usr/bin** :

```
cd /usr/bin
```

On va maintenant télécharger la version du fichier binaire de ueForth Linux :

- soit depuis la page d'accueil du site ESP32forth de Brad NELSON : https://esp32forth.appspot.com/ESP32forth.html
- soit depuis le dépôt de stockage eforth Google : https://eforth.storage.googleapis.com/releases/archive.html

Dans la liste des fichiers proposés, copiez le lien web mentionnant linux :

```
https://eforth.storage.googleapis.com/releases/ueforth-7.0.7.15.linux
```

Sous Linux, tapez la commande wget :

```
sudo wget https://eforth.storage.googleapis.com/releases/ueforth-7.0.7.15.linux
```

Le téléchargement va déposer automatiquement le fichier dans le dossier **usr/bin** précédemment sélectionné. Si vous avez repris le lien ci-avant, vous vous retrouvez avec un fichier nommé **ueforth-7.0.7.15.linux** dans ce dossier.

On renomme ce fichier avec la commande my:

```
sudo mv ueforth-7.0.7.15.linux ueforth
```

On vérifie que tout s'est bien déroulé avec une simple commande dir ue*. On doit voir la présence de notre fichier **ueforth**.

Il nous reste une dernière manipulation à effectuer, rendre ce fichier exécutable par le système Linux :

```
sudo chmod 755 ueforth
```

Et c'est fini! eForth Linux est maintenant utilisable depuis n'importe quel répertoire Linux.

Lancer eForth Linux

Pour lancer **eForth** au démarrage de **Linux** :

```
ueforth
```

eForth Linux démarre aussitôt :

```
mpmp9@LAPTOP-KL9DD6UH:~$ ueforth
uEforth v7.0.7.15 - rev 564a8fc68b545ebeb3ab
Forth dictionary: 10207640 free + 81276 used = 10288916 total (99% free)

3 x Forth stacks: 65536 bytes each
ok
--> |
```

Figure 2: eForth Linux est actif

Vous pouvez maintenant tester eForth et programmer vos premières applications en langage FORTH.

ATTENTION : cette version eForth gère les entiers au format 64 bits. C'est facile à vérifier :

```
cell . \ display : 8
```

Soit une dimension de 8 octets pour les entiers. Cet avertissement est indispensable si vous reprenez du code FORTH écrit pour des versions 16 ou 32 bits.

Bonne programmation.

Un vrai FORTH 64 bits avec eForth Linux

eForth Linux est un vrai FORTH 64 bits. Qu'est-ce que ça signifie ?

Le langage FORTH privilégie la manipulation de valeurs entières. Ces valeurs peuvent être des valeurs littérales, des adresses mémoires, des contenus de registres...

Les valeurs sur la pile de données

Au démarrage de eForth Linux, l'interpréteur FORTH est disponible. Si vous entrez n'importe quel nombre, il sera déposé sur la pile sous sa forme d'entier 64 bits :

```
35
```

Si on empile une autre valeur, elle sera également empilée. La valeur précédente sera repoussée vers le bas d'une position :

```
45
```

Pour faire la somme de ces deux valeurs, on utilise un mot, ici + :

```
+
```

Nos deux valeurs entières 64 bits sont additionnées et le résultat est déposé sur la pile. Pour afficher ce résultat, on utilisera le mot .:

```
. \ affiche 80
```

En langage FORTH, on peut concentrer toutes ces opérations en une seule ligne:

```
35 45 + . \ display 80
```

Contrairement au langage C, on ne définit pas de type int8 ou int16 ou int32 ou int64.

Avec eForth Linux, un caractère ASCII sera désigné par un entier 64 bits, mais dont la valeur sera bornée [32..256]. Exemple :

```
67 emit \ display C
```

Les valeurs en mémoire

eForth Linux permet de définir des constantes, des variables. Leur contenu sera toujours au format 64 bits. Mais il est des situations où ça ne nous arrange pas forcément. Prenons un exemple simple, définir un alphabet morse. Nous n'avons besoin que de quelques octets :

- un pour définir le nombre de signes du code morse
- un ou plusieurs octets pour chaque lettre du code morse

```
create mA ( -- addr )
   2 c,
   char . c,   char - c,
```

```
create mB ( -- addr )
   4 c,
   char - c,   char . c,   char . c,

create mC ( -- addr )
   4 c,
   char - c,   char . c,   char . c,
```

Ici, nous définissons seulement 3 mots, mA, mB et mC. Dans chaque mot, on stocke plusieurs octets. La question est: comment va-t-on récupérer les informations dans ces mots?

L'exécution d'un de ces mots dépose une valeur 64 bits, valeur qui correspond à l'adresse mémoire où on a stocké nos informations morse. C'est le mot c@ qui va nous servir à extraire le code morse de chaque lettre :

```
mA c@ . \ affiche 2
mB c@ . \ affiche 4
```

Le premier octet extrait ainsi va nous servir à gérer une boucle pour afficher le code morse d'une lettre :

```
: .morse ( addr -- )
   dup 1+ swap c@ 0 do
      dup i + c@ emit
   loop
   drop
;
mA .morse \ affiche .-
mB .morse \ affiche -...
mC .morse \ affiche -.-.
```

Il existe plein d'exemples certainement plus élégants. Ici, c'est pour montrer une manière de manipuler des valeurs 8 bits, nos octets, alors qu'on exploite ces octets sur une pile 64 bits.

Traitement par mots selon taille ou type des données

Dans tous les autres langages, on a un mot générique, genre echo (en PHP) qui affiche n'importe quel type de donnée. Que ce soit entier, réel, chaîne de caractères, on utilise toujours le même mot. Exemple en langage PHP:

```
$bread = "Pain cuit";
$price = 2.30;
echo $bread . " : " . $price;
// affiche Pain cuit: 2.30
```

Pour tous les programmeurs, cette manière de faire est LA NORME! Alors comment ferait FORTH pour cet exemple en PHP?

```
: pain s" Pain cuit" ;
: prix s" 2.30" ;
pain type s" : " type prix type
\ affiche Pain cuit: 2.30
```

Ici, le mot type nous indique qu'on vient de traiter une chaîne de caractères.

Là où PHP (ou n'importe quel autre langage) a une fonction générique et un analyseur syntaxique, FORTH compense avec un type de donnée unique, mais des méthodes de traitement adaptées qui nous informent sur la nature des données traitées.

Voici un cas absolument trivial pour FORTH, afficher un nombre de secondes au format HH:MM:SS:

```
: :##
    # 6 base !
    # decimal
    [char] : hold
;
: .hms ( n -- )
    <# :## :## # # > type
;
4225 .hms \ display: 01:10:25
```

J'adore cet exemple, car, à ce jour, **AUCUN AUTRE LANGAGE DE PROGRAMMATION** n'est capable de réaliser cette conversion HH:MM:SS de manière aussi élégante et concise.

Vous l'avez compris, le secret de FORTH est dans son vocabulaire.

Conclusion

FORTH n'a pas de typage de données. Toutes les données transitent par une pile de données. Chaque position dans la pile est TOUJOURS un entier 64 bits!

C'est tout ce qu'il y a à savoir.

Les puristes de langages hyper structurés et verbeux, tels C ou Java, crieront certainement à l'hérésie. Et là, je me permettrai de leur répondre : pourquoi avez-vous besoin de typer vos données ?

Car, c'est dans cette simplicité que réside la puissance de FORTH: une seule pile de données avec un format non typé et des opérations très simples.

Et je vais vous montrer ce que bien d'autres langages de programmation ne savent pas faire, définir de nouveaux mots de définition :

```
dup i + c@ emit
    loop
    drop space
;
2 morse: mA     char . c,     char - c,
4 morse: mB     char - c,     char . c,     char . c,
4 morse: mC     char - c,     char . c,     char . c,
mA mB mC     \ display     .- -... -.-.
```

Ici, le mot morse: est devenu un mot de définition, au même titre que constant ou variable...

Car FORTH est plus qu'un langage de programmation. C'est un méta-langage, c'est à dire un langage pour construire votre propre langage de programmation....

Edition et gestion fichiers sources pour eForth Linux

Comme pour la très grande majorité des langages de programmation, les fichiers sources écrits en langage FORTH sont au format texte simple. L'extension des fichiers en langage FORTH est libre :

- **txt** extension générique pour tous les fichiers texte ;
- forth utilisé par certains programmeurs FORTH ;
- fth forme compressée pour FORTH;
- 4th autre forme compressée pour FORTH;
- fs notre extension préférée...

Les éditeurs de fichiers texte




Figure 3: édition du fichier autoexec.fs avec gedit sous Linux

Sous Linux, l'éditeur de fichiers gedit est le plus simple :

Si vous utilisez une extension de fichier personnalisé, comme **fs**, pour vos fichiers source en langage FORTH, Linux reconnaîtra ces fichiers comme textes simples.

Stockage sur GitHub

Le site web **GitHub**² est, avec **SourceForge**³, un des meilleurs endroits

pour stocker ses fichiers sources.



3 https://sourceforge.net/




Figure 4: stockage des fichiers sur Github

Sur GitHub, vous pouvez mettre un dossier de travail en commun avec d'autres développeurs et gérer des projets complexes. L'éditeur Netbeans peut se connecter au projet et vous permet de transmettre ou récupérer des modifications de fichiers.




Figure 5: accès à une branche dans un projet

Sur **GitHub**, vous pouvez gérer des embranchements de projets (*fork*). Vous pouvez aussi rendre confidentiel certaines parties de vos projets. Ci-dessus les branches dans les projets de flagxor/ueforth :

Editer des fichiers pour eForth Linux depuis Windows

Si vous avez installé une version Linux qui s'exécute dans l'environnement WSL2, il est parfaitement possible d'éditer les fichiers sources Linux depuis Windows :

- lancez Ubuntu depuis Windows
- une fois Ubuntu actif, sortez le pointeur de souris de la fenêtre WSL. Vous revenez ainsi dans l'environnement Windows. Ouvrez le gestionnaire de fichiers Windows.
- dans le volet de gauche, cliquez sur *Linux*;




Figure 6: accès aux fichiers Linux depuis Windows

• dans le volet principal, cliquez sur la version Linux, ici *Ubuntu*;




Figure 7: les fichiers Linux visibles depuis Windows

- accédez au dossier eForth : dossier home → Utilisateur → ueforth →
- sélectionnez le fichier à éditer. Pour l'exemple, on va ouvrir autoexec.fs;

si vous utilisez un IDE, comme Netbeans, voici comment paramétrer cet IDE pour y intégrer vos projets de développement eForth Linux.

Création et gestion de projets FORTH avec Netbeans

En pré-requis, vous devez installer Netbeans. Lien pour téléchargement et installation : https://netbeans.apache.org/front/main/

Netbeans peut s'installer sous Windows ou Linux. Pour ma part, ayant déjà Netbeans installé sous Windows, je ne vais pas surcharger ma machine en installant une version Linux. En conséquence, les explications qui suivent concernent la gestion d'un projet eForth Linux via WSL2 depuis Windows.

Créer un projet eForth avec Netbeans

Là, également un pré-requis :

- ueforth Linux est installé dans Linux via WSL2 Windows.
- Les fichiers sources sont dans un dossier Windows :
 Linux → Ubuntu → home → userName → ueforth
 où userName est le nom d'utilisateur défini à l'installation de Linux
- tous les fichiers source eForth Linux sont enregistrés dans le répertoire ueforth

Lancez Netbeans. Pour créer un nouveau projet Netbeans :

cliquez sur File → sélectionnez New Project...

 dans la fenêtre New Project, sélectionnez Catégories : PHP et dans Projects : PHP Application with Existing Sources




Figure 8: création projet PHP

- cliquez sur Next >
- Dans le champ Name and Location → Sources Folder, saisissez le chemin des fichiers source eForth Linux

Pour retrouver le bon chemin du dossier **ueforth** depuis Windows, lancez l'explorateur de fichiers. Dans la partie inférieure droite, cliquez sur **Ubuntu**. Ensuite cliquez sur les dossiers :

home → *userName* → ueforth

Dans le bandeau de navigation, en haut, vous devez retrouver le chemin du dossier ueforth. Posez le pointeur de souris dans ce bandeau. Copiez le chemin :




Figure 9: copie du chemin vers ueforth sous Linux

Collez ce chemin dans le champ de Netbeans décrit plus haut. Terminez la création du nouveau projet dans Netbeans. Vous pouvez maintenant retrouver tous les fichiers de votre projet dans Netbeans :




Figure 10: le nouveau projet est opérationnel

Maintenant, toute édition, création, modification ou suppression de fichier depuis Netbeans est immédiatement répercuté dans le dossier de votre projet **ueforth** sous Linux.

Quelques bonnes pratiques

La première bonne pratique consiste à bien nommer ses fichiers et dossiers de travail. Vous développez pour eforth, donc restez dans le dossier nommé **ueforth**.

Pour les essais divers, créez dans ce dossier un sous-dossier sandbox (bac à sable).

Pour les projets bien construits, créez un dossier par projet. Par exemple, vous voulez développer un jeu, créez un sous-dossier **myGame**.

Si vous avez des scripts d'usage général, créez un sous-dossiet **tools**. Si vous utilisez un fichier de ce dossier **tools** dans un projet, copiez et collez ce fichier dans le dossier de ce projet. Ceci évitera qu'une modification d'un fichier dans **tools** ne perturbe ensuite votre projet.

Pour les essais FORTH sans but précis, mettez-les dans un dossier **sandbox**.

La seconde bonne pratique consiste à répartir le code source d'un projet dans plusieurs fichiers :

- config.fs pour stocker les paramètres du projet ;
- répertoire documentation pour stocker des fichiers dans le format de votre choix, en rapport avec la documentation du projet;




Figure 11: exemple de nommage de fichiers source Forth

• **myApp.fs** pour les définitions de votre projet. Choisissez un nom de fichier assez explicite. Par exemple, pour gérer votre jeu, prenez le nom **game-commands.fs**.

Exécution du contenu d'un fichier par eForth Linux

Depuis eForth Linux, l'exécution du contenu d'un fichier source s'effectue très simplement en utilisant le mot include suivi du nom de fichier :

```
include autoexec.fs
```

exécute le contenu du fichier autoexec.fs.

Si le fichier à lire est dans un sous-dossier, on fera précéder le nom du fichier par le nom du dossier. Exemple pour lancer **main.fs** dans le sous-dossier **myGame** :

```
cd mygame include main.fs
```

Si vous avez correctement installé **ueforth**, son lancement peut être suivi du nom du ficher source à exécuter. Sous Linux :

```
cd ueforth
ueforth UTF8.fs
```

Figure 12: exécution d'un fichier au lancement de ueforth

Linux enregistre toutes les commandes système, même après arrêt du PC et redémmarage. Il est donc très facile de recommencer un traitement de projet en quelques appuis de touche *flèche vers le haut*.

En résumé, sous réserve d'avoir une version Linux accessible depuis **Windows WSL2**, vous éditez les fichiers source avec Netbeans depuis Windows. Et vous traitez les fichiers d'un projet depuis Linux.

Si vous êtes dans un environnement *full Linux*, les manœuvres ne sont pas très différentes. Pour lancer ueforth, vous devrez ouvrir une fenêtre de commandes sous Linux.

Le système de fichiers Linux

eForth Linux intègre les composants essentiels pour accéder aux fichiers du système Linux.

Pour compiler le contenu d'un fichier source, ici le fichier **dumpTool.fs** dans le dossier **tools**, édité par **gedit**, taper :

```
include /tools/dumpTool.fs
```

Le mot include est un mot du dictionnaire eForth.

Pour voir la liste des fichiers Linux, utilisez le mot ls:

```
ls \ display :
...
autoexec.fs
blocks.fb
ueforth.bin
tools
ok
```

Ici, on voit le dossier **tools**. Eforth Linux n'utilise pas de coloration syntaxique comme le fait Linux. Pour voir le contenu de ce sous-dossier **tools**, taper :

```
ls tools \ display :
ls tools
.
..
dumpTool.fs
```

Il n'y a pas d'option de filtrage des noms de fichiers ou de pseudo-répertoires.

Manipulation des fichiers

Pour effacer intégralement un fichier, utiliser le mot rm suivi du nom de fichier à supprimer. Ici on souhaite effacer le fichier myTest.fs qui a été créé et ne sert plus :

```
rm myTest.fs \ display :
   ok
```

Pour renommer un fichier, utilisez le mot mv. Par exemple, on veut renommer un fichier myTest.txt:

```
mv myTest.txt myTest.fs
ls \ display :
...
autoexec.fs
blocks.fb
```

```
myTest.fs
tools
```

Pour copier un fichier, utilisez le mot cp:

```
cp myTest.fs testColors.fs
ls \ display :
...
autoexec.fs
blocks.fb
myTest.fs
testColors.fs
tools
```

Pour voir le contenu d'un fichier, utilisez le mot cat:

```
cat autoexec.fs
\ affiche contenu de autoexec.fs
```

Pour enregistrer le contenu d'une chaîne dans un fichier, on enregistre le contenu de la chaîne avec dump-file :

```
r| ." Insère mon texte dans myTest" | s" myTest.fs" dump-file
```

On ne s'étendra pas sur ces manipulations qui peuvent aussi bien s'effectuer depuis Linux ou un éditeur de texte source.

Organiser et compiler ses fichiers avec eForth Linux

Nous allons voir comment gérer des fichiers pour une application en cours de mise au point avec eForth Linux.

Il est convenu que tous les fichiers utilisés sont au format texte ASCII.

Les explications qui suivent ne sont données qu'à titre de conseils. Ils sont issus d'une certaine expérience et ont pour but de faciliter le développement de grosses applications avec eForth Linux.

Tous les fichiers sources de votre projet sont sur votre ordinateur dans l'environnement Linux. Il est conseillé d'avoir un sous-dossier dédié à ce projet. Par exemple, vous travaillez sur un jeu nommé rubik, vous créez donc un répertoire nommé **rubik**.

Concernant les extensions des noms de fichiers, nous conseillons d'utiliser l'extension fs.

L'édition des fichiers sur ordinateur est réalisée avec n'importe quel éditeur de fichiers texte, **gedit** sous Linux.

Dans ces fichiers sources, ne pas utiliser de caractère non inclus dans les caractères du code ASCII. Certains codes étendus peuvent perturber la compilation des programmes.

Organiser ses fichiers

Dans la suite, tous nos fichiers auront l'extension fs.

Partons de notre répertoire rubik sur notre ordinateur.

Le premier fichier que nous allons créer dans ce répertoire sera le fichier **main.fs**. Ce fichier contiendra les appels à chargement de tous les autres fichiers de notre application en cours de développement.

Exemple de contenu de notre fichier main.fs:

```
\ RUBIK game main file s" config.fs" included
```

En phase de développement, le contenu de ce fichier **main.fs** sera chargé depuis un fichier **RUBIK.fs** placé dans le même dossier que eForth et contenant ceci :

```
cd rubik
s" main.fs" included
```

Ceci provoque l'exécution du contenu de notre fichier **main.fs**. Le chargement des autres fichiers sera exécuté depuis ce fichier **main.fs**. Ici on exécute le chargement du fichier **config.fs** dont voici un extrait:

```
0 value MAX_DEPTH
3 constant CUBE_SIZE
```

Dans ce fichier **config.fs** on mettra toutes les valeurs constantes et divers paramètres globaux utilisés par les autres fichiers.




Figure 13: enchaînement des fichiers du projet RUBIK

Il est conseillé de mettre tous les fichiers d'un même projet dans le dossier de ce projet, ici **rubik** pour notre exemple.

Enchaînement des fichiers

Chaque fichier peut faire appel à un fichier avec le mot included. Voici un exemple de hiérarchie de fichiers ainsi inclus :




Figure 14: enchaînement de fichiers

Ici, eForth appelle un premier fichier. Même si c'est faisable, il est déconseillé de réaliser des enchaînements en cascade. Préférez une succession de chargement de fichiers depuis **main.fs.** Exemple :

```
DEFINED? --tempusFugit [if] forget --tempusFugit [then]
create --tempusFugit
s" strings.fs" included
s" RTClock.fs" included
s" clepsydra.fs" included
s" config.fs" included
s" dispTools.fs" included
```

Dans cette succession de fichiers, on utilise le fichier **strings.fs**. C'est un fichier dit *outil*. C'est la copie d'un fichier d'usage assez général et dont le contenu étend le dictionnaire FORTH.

En travaillant avec une copie du fichier d'origine, on peut y apporter des corrections ou des améliorations sans risquer d'altérer le fonctionnement du code dans le fichier d'origine. Si ces modifications sont consolidées, on peut les transférer dans le fichier d'origine.

Pour chaque fichier de code source FORTH, datez les versions. Ça vous permettra de retrouver la chronologie des modifications de code.

Conclusion

Les fichiers enregistrés dans le système Linux sont disponibles de manière permanente. Si vous accédez à une version Linux dans un système de gestion WSL2 depuis Windows, ces fichiers seront aussi accessibles au système de fichiers Windows.

Commentaires et mise au point

Il n'existe pas d'IDE⁴ pour gérer et présenter le code écrit en langage FORTH de manière structurée. Au pire, vous utilisez un éditeur de texte ASCII, au mieux un vrai IDE et des fichiers texte :

- edit ou wordpad sous Windows
- edit sous Linux
- PsPad sous windows
- Netbeans sous Windows ou Linux...

Voici un extrait de code qui pourrait être écrit par un débutant :

```
: inGrid? { n gridPos -- fl } 0 { fl } gridPos getGridAddr for aft getNumber n = if -1 to fl then then next drop fl ;
```

Ce code sera parfaitement compilé par eForth Linux. Mais restera-t-il compréhensible dans le futur s'il faut le modifier ou le réutiliser dans une autre application ?

Ecrire un code FORTH lisible

Commençons par le nomage du mot à définir, ici inGrid?. Eforth Linux permet d'écrire des noms de mots très longs. La taille des mots définis n'a aucune influence sur les performances de l'application finale. On dispose donc d'une certaine liberté pour écrire ces mots :

- à la manière de la programmation objet en javaScript: grid.test.number
- à la manière CamelCoding gridTestNumber
- pour programmeur voulant un code très compréhensible is-number-in-the-grid
- programmeur qui aime le code concis gtn?

Il n'y a pas de règle. L'essentiel est que vous puissez facilement relire votre code FORTH. Cependant, les programmeurs informatique en langage FORTH ont certaines habitudes :

- constantes en caractères majuscules LOTTO NUMBERS IN GRID
- mot de définition d'autres mots lottoNumber: mot suivi de deux points;
- mot de transformation d'adresse >date, ici le paramètre d'adresse est incrémenté d'une certaine valeur pour pointer sur la donnée adéquate;
- mot de stockage mémoire date@ ou date!

⁴ Integrated Development Environment = Environnement de Développement Intégré

• Mot d'affichage de donnée .date

Et qu'en est-il du nommage des mots FORTH dans une langue autre qu'en anglais ? Là encore, une seule règle : **liberté totale**! Attention cependant, eForth Linux n'accepte pas les noms écrits dans des alphabets différents de l'alphabet latin. Vous pouvez cependant utiliser ces alphabets pour les commentaires :

```
: .date \ Плакат сегодняшней даты ....code... ;
```

ou

```
: .date \ 海報今天的日期 .....code... ;
```

Indentation du code source

Que le code soit sur deux lignes, dix lignes ou plus, ça n'a aucun effet sur les performances du code une fois compilé. Donc, autant indenter son code de manière structurée :

- une ligne par mot de structure de contrôle **if else then, begin while repeat...**Pour le mot if, on peut de faire précéder du test logique qu'il traitera ;
- une ligne par exécution d'un mot prédéfini, précédé le cas échéant des paramètres de ce mot.

Exemple:

Si le code traité dans une structure de contrôle est peu fourni, le code FORTH peut être compacté :

```
: inGrid? { n gridPos -- fl }
    0 { fl } gridPos getGridAddr
    for aft
        getNumber n =
        if -1 to fl then
```

```
then
next
drop fl
;
```

C'est d'ailleurs souvent le cas avec des structures case of endof endcase;

```
: socketError ( -- )
   errno dup
   case
              ." No such file "
       2 of
                                                  endof
      5 of
               ." I/O error "
                                                  endof
       9 of
               ." Bad file number "
                                                  endof
      22 of
               ." Invalid argument "
                                                  endof
   endcase
   . quit
```

Les commentaires

Comme tout langage de programmation, le langage FORTH permet le rajout de commentaires dans le code source. Le rajout de commentaires n'a aucune conséquence sur les performances de l'application après compilation du code source.

En langage FORTH, nous disposons de deux mots pour délimiter des commentaires :

- le mot (suivi impérativement d'au moins un caractère espace. Ce commentaire est achevé par le caractère) ;
- le mot \ suivi impérativement d'au moins un caractère espace. Ce mot est suivi d'un commentaire de taille quelconque entre ce mot et la fin de la ligne.

Le mot (est largement utilisé pour les commentaires de pile. Exemples :

```
dup ( n - n n )
swap ( n1 n2 - n2 n1 )
drop ( n -- )
emit ( c -- )
```

Les commentaires de pile

Comme nous venons de le voir, ils sont marqués par (et). Leur contenu n'a aucune action sur le code FORTH en compilation ou en exécution. On peut donc mettre n'importe quoi entre (et). Pour ce qui concerne les commentaires de pile, on restera très concis. Le signe -- symbolise l'action d'un mot FORTH. Les indications figurant avant -- correspondent aux données déposées sur la pile de données avant l'exécution du mot. Les indications figurant après -- correspondent aux données laissées sur la pile de données après exécution du mot. Exemples :

- words (--) signifie que ce mot ne traite aucune donnée sur la pile de données ;
- emit (c --) signifie que ce mot traite une donnée en entrée et ne laisse rien sur la pile de données ;
- **bl** (-- **32**) signifie que ce mot ne traite pas de donnée en entrée et laisse la valeur décimale 32 sur la pile de données ;

Il n'y a aucune limitation sur le nombre de données traitées avant ou après exécution du mot. Pour rappel, les indications entre (et) sont seulement là pour information.

Signification des paramètres de pile en commentaires

Pour commencer, une petite mise au point très importante s'impose. Il s'agit de la taille des données en pile. Avec eForth Linux, les données de pile occupent 8 octets. Ce sont donc des entiers au format 64 bits. Alors on met quoi sur la pile de données ? Avec eForth Linux, ce seront **TOUJOURS DES DONNEES 64 BITS**! Un exemple avec le mot c!:

```
create myDelemiter
    0 c,
64 myDelimiter c! ( c addr -- )
```

Ici, le paramètre c indique qu'on empile une valeur entière au format 64 bits, mais dont la valeur sera toujours comprise dans l'intervale [0..255].

Le paramètre standard est toujours n. S'il y a plusieurs entiers, on les numérotera : n1 n2 n3, etc.

On aurait donc pu écrire l'exemple précédent comme ceci :

```
create myDelemiter
    0 c,
64 myDelimiter c! ( n1 n2 -- )
```

Mais c'est nettement moins explicite que la version précédente. Voici quelques symboles que vous serez amené à voir au fil des codes sources :

- addr indique une adresse mémoire littérale ou délivrée par une variable ;
- **c** indique une valeur 8 bits dans l'intervalle [0..255]
- d indique une valeur double précision.
 Non utilisé avec eForth Linux qui est déjà au format 64 bits ;
- fl indique une valeur booléenne, 0 ou non zéro ;
- **n** indique un entier. Entier signé 64 bits pour eForth Linux;
- str indique une chaîne de caractère. Équivaut à addr len --
- u indique un entier non signé

Rien n'interdit d'être un peu plus explicite :

```
: SQUARE ( n -- n-exp2 )
dup *
;
```

Commentaires des mots de définition de mots

Les mots de définition utilisent **create** et **does**>. Pour ces mots, il est conseillé d'écrire les commentaires de pile de cette manière :

Ici, le commentaire est partagé en deux parties par le caractère | :

- à gauche, la partie action quand le mot de définition est exécuté, préfixé par comp:
- à droite la partie action du mot qui sera défini, préfixé par exec:

Au risque d'insister, ceci n'est pas un standard. Ce sont seulement des recommandations.

Les commentaires textuels

Ils sont inqués par le mot \ suivi obligatoirement par au moins un caractère espace et du texte explicatif :

```
\ store at <WORD> addr length of datas compiled beetween
\ <WORD> and here
: ;endStream ( addr-var len ---)
    dup 1+ here
    swap - \ calculate cdata length
    \ store c in first byte of word defined by streamCreate:
    swap c!
;
```

Ces commentaires peuvent être écrits dans n'importe quel alphabet supporté par votre éditeur de code source :

```
\ 儲存在 <WORD> addr 之間編譯的資料長度
\ <WORD> 和這裡
: ;endStream ( addr-var len ---)
    dup 1+ here
    swap - \ 計算cdata 長度
    \ 將 c 儲存在由 StreamCreate 定義的字的第一個位元組中:
```

```
swap c!
```

Commentaire en début de code source

Avec une pratique de programmation intensive, on se retrouve rapidement avec des centaines, voire des milliers de fichiers source. Pour éviter des erreurs de choix de fichiers, il est fortement conseillé de marquer le début de chaque fichier source avec un commentaire :

```
\ ************************
\ Manage commands for OLED SSD1306 128x32 display
   Filename: SSD10306commands.fs
   Date:
               21 may 2023
١
   Updated: 21 may 2023
   File Version: 1.0
               ESP32-WROOM-32
  MCU:
   Forth:
               ESP32forth all versions 7.x++
   Copyright: Marc PETREMANN
   Author:
               Marc PETREMANN
    GNU General Public License
\ **************
```

Toutes ces informations sont à votre libre choix. Elles peuvent devenir très utiles quand on revient des mois ou des années plus tard sur le contenu d'un fichier.

Pour conclure, n'hésitez pas à commenter et indenter vos fichiers sources en langage FORTH.

Outils de diagnostic et mise au point

Le premier outil concerne l'alerte de compilation ou d'interprétation :

```
3 5 25 --> : TEST ( ---)
ok
3 5 25 --> [ HEX ] ASCII A DDUP \ DDUP don't exist
```

Ici, le mot **DDUP** n'existe pas. Toute compilation après cette erreur sera vouée à l'échec.

Le décompilateur

Dans un compilateur conventionnel, le code source est transformé en code exécutable contenant les adresses de référence à une bibliothèque équipant le compilateur. Pour disposer d'un code exécutable , il faut linker le code objet. A aucun moment le programmeur ne peut avoir accès au code exécutable contenu dans ses bibliothèque avec les seules ressources du compilateur.

Avec eForth Linux, le développeur peut décompiler ses définitions. Pour décompiler un mot, il suffit de taper see suivi du mot à décompiler :

```
: C>F ( ØC --- ØF) \ Conversion Celsius in Fahrenheit
    9 5 */ 32 +
;
see c>f
\ display:
: C>F
    9 5 */ 32 +
;
```

Beaucoup de mots du dictionnaire FORTH de eForth Linux peuvent être décompilés. La décompilation de vos mots permet de détecter d'éventuelles erreurs de compilation.

Dump mémoire

Parfois, il est souhaitable de pouvoir voir les valeurs qui sont en mémoire. Le mot dump accepte deux paramètres: l'adresse de départ en mémoire et le nombre d'octets à visualiser :

```
create myDATAS 01 c, 02 c, 03 c, 04 c,
hex
myDATAS 4 dump \ displays :
3FFEE4EC 01 02 03 04
```

Moniteur de pile

Le contenu de la pile de données peut être affiché à tout moment grâce au mot .s. Voici la définition du mot .DEBUG qui exploite .s :

```
variable debugStack
: debugOn ( -- )
    -1 debugStack !
;
: debugOff ( -- )
    0 debugStack !
;
: .DEBUG
    debugStack @
    if
        cr ." STACK: " .s
        key drop
    then
;
```

Pour exploiter . DEBUG, il suffit de l'insérer dans un endroit stratégique du mot à mettre au point :

```
\ example of use:
```

```
: myTEST

128 32 do

i .DEBUG

emit

loop

;
```

Ici, on va afficher le contenu de la pile de données après exécution du mot i dans notre boucle do loop. On active la mise au point et on exécute myTEST:

```
debugOn
myTest
\ displays:
\ STACK: <1> 32
\ 2
\ STACK: <1> 33
\ 3
\ STACK: <1> 34
\ 4
\ STACK: <1> 35
\ 5
\ STACK: <1> 36
\ 6
\ STACK: <1> 37
\ 7
\ STACK: <1> 38
```

Quand la mise au point est activée par debugon, chaque affichage du contenu de la pile de données met en pause notre boucle do loop. Exécuter debugoff pour que le mot myTEST s'exécute normalement.

Dictionnaire / Pile / Variables / Constantes

Étendre le dictionnaire

Forth appartient à la classe des langages d'interprétation tissés. Cela signifie qu'il peut interpréter les commandes tapées sur la console, ainsi que compiler de nouveaux sousprogrammes et programmes.

Le compilateur Forth fait partie du langage et des mots spéciaux sont utilisés pour créer de nouvelles entrées de dictionnaire (c'est-à-dire des mots). Les plus importants sont : (commencer une nouvelle définition) et ; (termine la définition). Essayons ceci en tapant:

```
: *+ * + ;
```

Ce qui s'est passé? L'action de : est de créer une nouvelle entrée de dictionnaire nommée *+ et passer du mode interprétation au mode compilation. En mode compilation, l'interpréteur recherche les mots et, plutôt que de les exécuter, installe des pointeurs vers leur code. Si le texte est un nombre, au lieu de le pousser sur la pile, eForth Linux construit le nombre dans le dictionnaire l'espace alloué pour le nouveau mot, suivant le code spécial qui met le numéro stocké sur la pile chaque fois que le mot est exécuté. L'action d'exécution de *+ est donc d'exécuter séquentiellement les mots définis précédemment * et +.

Le mot ; est spécial. C'est un mot immédiat et il est toujours exécuté, même si le le système est en mode compilation. Ce que fait ; est double. Tout d'abord, il installe le code qui renvoie le contrôle au niveau externe suivant de l'interpréteur et, deuxièmement, il revient du mode compilation au mode interprétation.

Maintenant, essayez votre nouveau mot :

```
decimal 5 6 7 *+ . \ affiche 47 ok<#,ram>
```

Cet exemple illustre deux activités principales de travail dans Forth: ajouter un nouveau mot au dictionnaire, et l'essayer dès qu'il a été défini.

Gestion du dictionnaire

Le mot **forget** suivi du mot à supprimer enlèvera toutes les entrées de dictionnaire que vous avez faites depuis ce mot:

```
: test1 ;
: test2 ;
: test3 ;
forget test2 \ efface test2 et test3 du dictionnaire
```

Piles et notation polonaise inversée

Forth a une pile explicitement visible qui est utilisée pour passer des nombres entre les mots (commandes). Utiliser Forth efficacement vous oblige à penser en termes de pile. Cela peut être difficile au début, mais comme pour tout, cela devient beaucoup plus facile avec la pratique.

En FORTH, La pile est analogue à une pile de cartes avec des nombres écrits dessus. Les nombres sont toujours ajoutés au sommet de la pile et retirés du sommet de la pile. eForth Linux intègre deux piles: la pile de paramètres et la pile de retour, chacune composée d'un certain nombre de cellules pouvant contenir des nombres de 16 bits.

La ligne d'entrée FORTH:


```
decimal 2 5 73 -16
```

laisse la pile de paramètres dans l'état

Cellule	contenu	commentaire
0	-16	(TOS) Sommet pile
1	73	(NOS) Suivant dans la pile
2	5	
3	2	

Nous utiliserons généralement une numérotation relative à base zéro dans les structures de données Forth telles que piles, tableaux et tables. Notez que, lorsqu'une séquence de nombres est saisie comme celle-ci, le nombre le plus à droite devient *TOS* et le nombre le plus à gauche se trouve au bas de la pile.

Supposons que nous suivions la ligne d'entrée d'origine avec la ligne



Les opérations produiraient les opérations de pile successives:

Après les deux lignes, la console affiche :

```
decimal 2 5 73 -16 \ affiche: 2 5 73 -16 ok + - * . \ \ affiche: -104 ok
```

Notez que eForth Linux affiche commodément les éléments de la pile lors de l'interprétation de chaque ligne et que la valeur de -16 est affichée sous la forme d'entier non signé 64 bits. En outre, le mot . consomme la valeur de données -104, laissant la pile vide. Si nous exécutons . sur la pile maintenant vide, l'interpréteur externe abandonne avec une erreur de pointeur de pile STACK UNDERFLOW ERROR.

La notation de programmation où les opérandes apparaissent en premier, suivis du ou des opérateurs est appelée Notation polonaise inverse (RPN).

Manipulation de la pile de paramètres

Étant un système basé sur la pile, eForth Linux doit fournir des moyens de mettre des nombres sur la pile, pour les supprimer et réorganiser leur ordre. On a déjà vu qu'on peut mettre des nombres sur la pile simplement en les tapant. Nous pouvons également intégrer les nombres dans la définition d'un mot FORTH.

Le mot drop supprime un numéro du sommet de la pile mettant ainsi le suivant au sommet. Le mot swap échange les 2 premiers numéros. dup copie le nombre au sommet, poussant tout les autres numéros vers le bas. rot fait pivoter les 3 premiers nombres. Ces



actions sont présentées ci-dessous.

La pile de retour et ses utilisations

Lors de la compilation d'un nouveau mot, eForth Linux établit des liens entre le mot appelant et les mots définis précédemment qui doivent être invoqués par l'exécution du nouveau mot. Ce mécanisme de liaison, lors de l'exécution, utilise la pile de retour (rstack). L'adresse du mot suivant à invoquer est placé sur la pile de retour de sorte que, lorsque le mot courant est terminé en cours d'exécution, le système sait où passer au mot suivant. Comme les mots peuvent être imbriqués, il doit y avoir une pile de ces adresses de retour.

En plus de servir de réservoir d'adresses de retour, l'utilisateur peut également stocker et récupérer à partir de la pile de retour, mais cela doit être fait avec soin car la pile de retour est essentielle à l'exécution du programme. Si vous utilisez la pile de retour pour le

stockage temporaire, vous devez la remettre dans son état d'origine, sinon vous ferez probablement planter le système eForth Linux. Malgré le danger, il y a des moments où l'utilisation de pile de retour comme stockage temporaire peut rendre votre code moins complexe.

Pour stocker dans la pile, utilisez >r pour déplacer le sommet de la pile de paramètres vers le haut de la pile de retour. Pour récupérer une valeur, r> déplace la valeur supérieure de la pile de retour vers le sommet de la pile de paramètres. Pour supprimer simplement une valeur du haut de la pile, il y a le mot rdrop. Le mot r@ copie le haut de la pile de retour dans la pile de paramètres.

Utilisation de la mémoire

Dans eForth Linux, les nombres 64 bits sont extraits de la mémoire vers la pile par le mot @ (fetch) et stocké du sommet à la mémoire par le mot ! (store). @ attend une adresse sur la pile et remplace l'adresse par son contenu. ! attend un nombre et une adresse pour le stocker. Il place le numéro dans l'emplacement de mémoire référencé par l'adresse, consommant les deux paramètres dans le processus.

Les nombres non signés qui représentent des valeurs de 8 bits (octets) peuvent être placés dans des caractères de la taille d'un caractère. cellules de mémoire en utilisant c@ et c!.

```
create testVar
   cell allot

$f7 testVar c!

testVar c@ . \ affiche 247
```

Variables

Une variable est un emplacement nommé en mémoire qui peut stocker un nombre, tel que le résultat intermédiaire d'un calcul, hors de la pile. Par exemple:

```
variable x
```

crée un emplacement de stockage nommé, x, qui s'exécute en laissant l'adresse de son emplacement de stockage au sommet de la pile:

```
x . \ affiche l'adresse
```

Nous pouvons alors aller chercher ou stocker à cette adresse :

```
variable x
3 x !
x @ . \ affiche: 3
```

Constantes

Une constante est un nombre que vous ne voudriez pas changer pendant l'exécution d'un programme. Le résultat de l'exécution du mot associé à un constante est la valeur des données restant sur la pile.

```
\ définit les pins VSPI
19 constant VSPI_MISO
23 constant VSPI_MOSI
18 constant VSPI_SCLK
05 constant VSPI_CS

\ définit la fréquence du port SPI
4000000 constant SPI_FREQ

\ sélectionne le vocabulaire SPI
only FORTH SPI also

\ initialise le port SPI
: init.VSPI ( -- )
    VSPI_CS OUTPUT pinMode
    VSPI_SCLK VSPI_MISO VSPI_MOSI VSPI_CS SPI.begin
    SPI_FREQ SPI.setFrequency
;
```

Valeurs pseudo-constantes

Une valeur définie avec value est un type hybride de variable et constante. Nous définissons et initialisons une valeur et est invoquée comme nous le ferions pour une constante. On peut aussi changer une valeur comme on peut changer une variable.

```
decimal

13 value thirteen
thirteen . \ display: 13

47 to thirteen
thirteen . \ display: 47
```

Le mot to fonctionne également dans les définitions de mots, en remplaçant la valeur qui le suit par tout ce qui est actuellement au sommet de la pile. Vous devez faire attention à ce que to soit suivi d'une valeur définie par value et non d'autre chose.

Outils de base pour l'allocation de mémoire

Les mots **create** et **allot** sont les outils de base pour réserver un espace mémoire et y attacher une étiquette. Par exemple, la transcription suivante montre une nouvelle entrée de dictionnaire **graphic-array** :

```
%0000100 c,
%0001000 c,
%0010000 c,
%0100000 c,
%1000000 c,
```

Lorsqu'il est exécuté, le mot graphic-array poussera l'adresse de la première entrée.

Nous pouvons maintenant accéder à la mémoire allouée à **graphic-array** en utilisant les mots de récupération et de stockage expliqués plus tôt. Pour calculer l'adresse du troisième octet attribué à **graphic-array** on peut écrire **graphic-array** 2 +, en se rappelant que les indices commencent à 0.

```
30 graphic-array 2 + c!
graphic-array 2 + c@ . \ affiche 30
```

Les variables locales avec eForth Linux

Introduction

Le langage FORTH traite les données essentiellement par la pile de données. Ce mécanisme très simple offre une performance inégalée. A contrario, suivre le cheminement des données peut rapidement devenir complexe. Les variables locales offrent une alternative intéressante.

Le faux commentaire de pile

Si vous suivez les différents exemples FORTH, vous avez noté les commentaires de pile encadrés par (et). Exemple:

```
\ addition deux valeurs non signées, laisse sum et carry sur la pile
: um+ ( u1 u2 -- sum carry )
     \ ici la définition
;
```

Ici, le commentaire (u1 u2 -- sum carry) n'a absolument aucune action sur le reste du code FORTH. C'est un pur commentaire.

Quand on prépare une définition complexe, la solution est d'utiliser des variables locales encadrées par { et }. Exemple:

```
: 20VER { a b c d }
    a b c d a b
;
```

On définit quatre variables locales a b c et d.

Les mots { et } ressemblent aux mots (et) mais n'ont pas du tout le même effet. Les codes placés entre { et } sont des variables locales. Seule contrainte: ne pas utiliser de noms de variables qui pourraient être des mots FORTH du dictionnaire FORTH. On aurait aussi bien pu écrire notre exemple comme ceci:

```
: 20VER { varA varB varC varD }
    varA varB varC varD varA varB
;
```

Chaque variable va prendre la valeur de la donnée de pile dans l'ordre de leur dépôt sur la pile de données. ici, 1 va dans varA, 2 dans varB, etc..:

```
--> 1 2 3 4
ok
1 2 3 4 --> 2over
ok
1 2 3 4 1 2 -->
```

Notre faux commentaire de pile peut être complété comme ceci:

```
: 20VER { varA varB varC varD -- varA varB varC varD varA varB } .....
```

Les caractères qui suivent -- n'ont pas d'effet. Le seul intérêt est de rendre notre faux commentaire semblable à un vrai commentaire de pile.

Action sur les variables locales

Les variables locales agissent exactement comme des pseudo-variables définies par value. Exemple:

```
: 3x+1 { var -- sum }
    var 3 * 1 +
;
```

A le même effet que ceci:

```
0 value var
: 3x+1 ( var -- sum )
    to var
    var 3 * 1 +
;
```

Dans cet exemple, var est défini explicitement par value.

On affecte une valeur à une variable locale avec le mot to ou +to pour incrémenter le contenu d'une variable locale. Dans cet exemple, on rajoute une variable locale result initialisée à zéro dans le code de notre mot:

```
: a+bEXP2 { varA varB -- (a+b)EXP2 }
    0 { result }
    varA varA * to result
    varB varB * +to result
    varA varB * 2 * +to result
    result
    ;
```

Est-ce que ce n'est pas plus lisible que ceci?

```
: a+bEXP2 ( varA varB -- result )
    2dup
    * 2 * >r
    dup *
    swap dup * +
    r> +
    ;
```

Voici un dernier exemple, la définition du mot um+ qui additionne deux entiers non signés et laisse sur la pile de données la somme et la valeur de débordement de cette somme:

```
\ addition deux entiers non signés, laisse sum et carry sur la pile
```

```
: um+ { u1 u2 -- sum carry }
    0 { sum }
    cell for
        aft
            u1 $100 /mod to u1
            u2 $100 /mod to u2
            +
            cell 1- i - 8 * lshift +to sum
        then
    next
    sum
    u1 u2 + abs
;
```

Voici un exemple plus complexe, la réécriture de **DUMP** en exploitant des variables locales:

```
\ variables locales dans DUMP:
\ START_ADDR \ première addresse pour dump
\ END ADDR
                 \ dernière addresse pour dump
\ 0START ADDR
                \ première adresse pour la boucle dans dump
\ LINES
                 \ nombre de lignes pour la boucle dump
                \ base numérique courante
\ myBASE
internals
: dump ( start len -- )
   cr cr ." --addr--- "
    ." 00 01 02 03 04 05 06 07 08 09 0A 0B 0C 0D 0E 0F -----chars----"
   2dup + { END ADDR }
                                   \ store latest address to dump
   swap { START ADDR }
                                   \ store START address to dump
   START ADDR 16 / 16 * { OSTART ADDR } \ calc. addr for loop start
   16 / 1+ { LINES }
   base @ { myBASE }
                                   \ save current base
   hex
   \ outer loop
   LINES 0 do
       OSTART_ADDR i 16 * +
                               \ calc start address for current line
       cr <# # # # # [char] - hold # # # # #> type
       space space \ and display address
       \ first inner loop, display bytes
       16 0 do
           \ calculate real address
           OSTART ADDR j 16 * i + +
           ca@ <# # # > type space \ display byte in format: NN
       loop
        \ second inner loop, display chars
        16 0 do
           \ calculate real address
           OSTART ADDR j 16 * i + +
```

```
\ display char if code in interval 32-127
ca@ dup 32 < over 127 > or
    if drop [char] . emit
    else emit
    then
    loop
loop
myBASE base ! \ restore current base
cr cr
;
forth
```

L'emploi des variables locales simplifie considérablement la manipulation de données sur les piles. Le code est plus lisible. On remarquera qu'il n'est pas nécessaire de pré-déclarer ces variables locales, il suffit de les désigner au moment de les utiliser, par exemple: base @ { myBASE }.

ATTENTION: si vous utilisez des variables locales dans une définition, n'utilisez plus les mots >r et r>, sinon vous risquez de perturber la gestion des variables locales. Il suffit de regarder la décompilation de cette version de DUMP pour comprendre la raison de cet avertissement:

```
: dump cr cr s" --addr--- " type
s" 00 01 02 03 04 05 06 07 08 09 0A 0B 0C 0D 0E 0F -----chars----" type
2dup + >R SWAP >R -4 local@ 16 / 16 * >R 16 / 1+ >R base @ >R
hex -8 local@ 0 (do) -20 local@ R@ 16 * + cr
<# # # # 45 hold # # # # > type space space
16 0 (do) -28 local@ j 16 * R@ + + CA@ <# # # * > type space 1 (+loop)
0BRANCH rdrop rdrop space 16 0 (do) -28 local@ j 16 * R@ + + CA@ DUP 32 < OVER 127 > OR
0BRANCH DROP 46 emit BRANCH emit 1 (+loop) 0BRANCH rdrop rdrop 1 (+loop)
0BRANCH rdrop rdrop -4 local@ base ! cr cr rdrop rdrop rdrop rdrop rdrop;
```

Structures de données pour eForth Linux

Préambule

eForth Linux est une version 64 bits du langage FORTH. Ceux qui ont pratiqué FORTH depuis ses débuts ont programmé avec des versions 16 ou 32 bits. Cette taille de données est déterminée par la taille des éléments déposés sur la pile de données. Pour connaître la taille en octets des éléments, il faut exécuter le mot cell. Exécution de ce mot pour eForth Linux:

```
cell . \ affiche 8
```

La valeur 8 signifie que la taille des éléments déposés sur la pile de donnéees est de 8 octets, soit 8x8 bits = 64 bits.

Avec une version FORTH 16 bits, cell empilera la valeur 2. De même, si vous utilisez une version 32 bits, cell empilera la valeur 4.

Les tableaux en FORTH

Commençons par des structures assez simples: les tableaux. Nous n'aborderons que les tableaux à une ou deux dimensions.

Tableau de données 32 bits à une dimension

C'est le type de tableau le plus simple. Pour créer un tableau de ce type, on utilise le mot create suivi du nom du tableau à créer:

```
create temperatures
34 , 37 , 42 , 36 , 25 , 12 ,
```

Dans ce tableau, on stocke 6 valeurs: 34, 37....12. Pour récupérer une valeur, il suffit d'utiliser le mot @ en incrémentant l'adresse empilée par **temperatures** avec le décalage souhaité:

```
temperatures \ empile addr
0 cell * \ calcule décalage 0
+ \ ajout décalage à addr
0 . \ affiche 34

temperatures \ empile addr
1 cell * \ calcule décalage 1
+ \ ajout décalage à addr
0 . \ affiche 37
```

On peut factoriser le code d'accès à la valeur souhaitée en définissant un mot qui va calculer cette adresse:

```
: temp@ ( index -- value )
   cell * temperatures + @
;
0 temp@ . \ affiche 34
2 temp@ . \ affiche 42
```

Vous noterez que pour n valeurs stockées dans ce tableau, ici 6 valeurs, l'index d'accès doit toujours être dans l'intervalle [0..n-1].

Mots de définition de tableaux

Voici comment créer un mot de définition de tableaux d'entiers à une dimension:

```
: array ( comp: -- | exec: index -- addr )
    create
    does>
        swap cell * +
;
array myTemps
    21 ,    32 ,    45 ,    44 ,    28 ,    12 ,
0 myTemps @ . \ affiche 21
5 myTemps @ . \ affiche 12
```

Dans notre exemple, nous stockons 6 valeurs comprises entre 0 et 255. Il est aisé de créer une variante de array pour gérer nos données de manière plus compacte:

Avec cette variante, on stocke les mêmes valeurs dans quatre fois moins d'espace mémoire.

Gestion de structures complexes

eForth Linux dispose du vocabulaire structures. Le contenu de ce vocabulaire permet de définir des structures de données complexes.

Voici un exemple trivial de structure :

```
structures
```

```
struct YMDHMS
  ptr field >year
  ptr field >month
  ptr field >day
  ptr field >hour
  ptr field >min
  ptr field >sec
```

Ici, on définit la structure YMDHMS. Cette structure gère les pointeurs >year >month >day >hour >min et >sec.

Le mot YMDHMS a comme seule utilité d'initialiser et regrouper les pointeurs dans la structure complexe. Voici comment sont utilisés ces pointeurs:

```
create DateTime
   YMDHMS allot
2022 DateTime >year !
 03 DateTime >month !
 21 DateTime >day !
 22 DateTime >hour !
 36 DateTime >min !
 15 DateTime >sec !
: .date ( date -- )
   >r
   ." YEAR: " r@ >year
                      @ . cr
   ." MONTH: " r@ >month @ . cr
   ." DAY: " r@ >day @ . cr
   ." HH: " r@ >hour @ . cr
   r> drop
DateTime .date
```

On a défini le mot **DateTime** qui est un tableau simple de 6 cellules 64 bits consécutives. L'accès à chacune des cellules est réalisée par l'intermédiaire du pointeur correspondant. On peut redéfinir l'espace alloué de notre structure **YMDHMS** en utilisant le mot **i8** pour pointer des octets:

```
structures
struct cYMDHMS

ptr field >year

i8 field >month

i8 field >day

i8 field >hour

i8 field >min
```

```
i8 field >sec
create cDateTime
   cYMDHMS allot
2022 cDateTime >year !
 03 cDateTime >month c!
 21 cDateTime >day c!
 22 cDateTime >hour c!
 36 cDateTime >min c!
 15 cDateTime >sec c!
: .cDate ( date -- )
   >r
   ." YEAR: " r@ >year
                         @ . cr
   ." MONTH: " r@ >month c@ . cr
   ." DAY: " r@ >day
                        c@ . cr
   . "
       HH: " r@ >hour c@ . cr
        MM: " r@ >min
                        c@ . cr
   ." SS: " r@ >sec
                        c@ . cr
   r> drop
cDateTime .cDate \ affiche:
 YEAR: 2022
\ MONTH: 3
   DAY: 21
    HH: 22
   MM: 36
    SS: 15
```

Dans cette structure cYMDHMS, on a gardé l'année au format 64 bits et réduit toutes les autres valeurs à des entiers 8 bits. On constate, dans le code de .cDate, que l'utilisation des pointeurs permet un accès aisé à chaque élément de notre structure complexe....

Les nombres réels avec eForth Linux

Si on teste l'opération 1 3 / en langage FORTH, le résultat sera 0.

Ce n'est pas surprenant. De base, eForth Linux n'utilise que des nombres entiers 64 bits via la pile de données. Les nombres entiers offrent certains avantages :

- rapidité de traitement ;
- résultat de calculs sans risque de dérive en cas d'itérations ;
- conviennent à quasiment toutes les situations.

Même en calculs trigonométriques, on peut utiliser une table d'entiers. Il suffit de créer un tableau avec 90 valeurs, où chaque valeur correspond au sinus d'un angle, multiplié par 1000.

Mais les nombres entiers ont aussi des limites :

- résultats impossibles pour des calculs de division simple, comme notre exemple 1/3;
- nécessite des manipulations complexes pour appliquer des formules de physique.

Depuis la version 7.0.6.5, eForth Linux intègre des opérateurs traitant des nombres réels. Les nombres réels sont aussi dénommés nombres à virgule flottante.

Les réels avec eForth Linux

Afin de distinguer les nombres réels, il faut les terminer avec la lettre "e":

```
3 \ empile 3 sur la pile de données
3e \ empile 3 sur la pile des réels
5.21e f. \ affiche 5.210000
```

C'est le mot f. qui permet d'afficher un nombre réel situé au sommet de la pile des réels.

Precision des nombres réels avec eForth Linux

Le mot **set-precision** permet d'indiquer le nombre de décimales à afficher après le point décimal. Voyons ceci avec la constante **pi**:

```
pi f. \ affiche 3.141592
4 set-precision
pi f. \ affiche 3.1415
```

La précision limite de traitement des nombres réels avec eForth Linux est de six décimales :

```
12 set-precision
```

```
1.987654321e f. \ affiche 1.987654668777
```

Si on réduit la précision d'affichage des nombres réels en dessous de 6, les calculs seront quand même réalisés avec une précision à 6 décimales.

Constantes et variables réelles

Une constante réelle est définie avec le mot fconstant:

```
0.693147e fconstant ln2 \ logarithme naturel de 2
```

Une variable réelle est définie avec le mot fvariable:

```
fvariable intensity

170e 12e F/ intensity SF! \ I=P/U --- P=170w U=12V

intensity SF@ f. \ affiche 14.166669
```

ATTENTION: tous les nombres réels transitent par la **pile des nombres réels**. Dans le cas d'une variable réelle, seule l'adresse pointant sur la valeur réelle transite par la pile de données.

Le mot **SF!** enregistre une valeur réelle à l'adresse ou la variable pointée par son adresse mémoire. L'exécution d'une variable réelle dépose l'adresse mémoire sur la pile données classique.

Le mot **SF**@ empile la valeur réelle pointée par son adresse mémoire.

Opérateurs arithmétiques sur les réels

eForth Linux dispose de quatre opérateurs arithmétiques F+ F- F* F/:

eForth Linux dispose aussi de ces mots :

- 1/F calcule l'inverse d'un nombre réel;
- fsgrt calcule la racine carrée d'un nombre réel.

```
5e 1/F f. \ affiche 0.200000 1/5
5e fsqrt f. \ affiche 2.236068 sqrt(5)
```

Opérateurs mathématiques sur les réels

eForth Linux dispose de plusieurs opérateurs mathématiques :

- F** élève un réel r_val à la puissance r_exp
- FATAN2 calcule l'angle en radian à partir de la tangente.

- FCOS (r1 -- r2) Calcule le cosinus d'un angle exprimé en radians.
- FEXP (In-r -- r) calcule le réel correspondant à e EXP r
- FLN (r -- ln-r) calcule le logarithme naturel d'un nombre réel.
- FSIN (r1 -- r2) calcule le sinus d'un angle exprimé en radians.
- **FSINCOS** (r1 -- rcos rsin) calcule le cosinus et le sinus d'un angle exprimé en radians.

Quelques exemples:

```
2e 3e f** f. \ affiche 8.000000
2e 4e f** f. \ affiche 16.000000
10e 1.5e f** f. \ affiche 31.622776

4.605170e FEXP F. \ affiche 100.000018

pi 4e f/
FSINCOS f. f. \ affiche 0.707106 0.707106
pi 2e f/
FSINCOS f. f. \ affiche 0.000000 1.000000
```

Opérateurs logiques sur les réels

eForth Linux permet aussi d'effectuer des tests logiques sur les réels :

- FO< (r -- fl) teste si un nombre réel est inférieur à zéro.
- F0= (r -- fl) indique vrai si le réel est nul.
- f< (r1 r2 -- fl) fl est vrai si r1 < r2.
- f<= (r1 r2 -- fl) fl est vrai si r1 <= r2.
- f<> (r1 r2 -- fl) fl est vrai si r1 <> r2.
- f= (r1 r2 -- fl) fl est vrai si r1 = r2.
- f> (r1 r2 -- fl) fl est vrai si r1 > r2.
- f>= (r1 r2 -- fl) fl est vrai si r1 >= r2.

Transformations entiers ↔ réels

eForth Linux dispose de deux mots pour transformer des entiers en réels et inversement :

- F>S (r -- n) convertit un réel en entier. Laisse sur la pile de données la partie entière si le réel a des parties décimales.
- S>F (n -- r: r) convertit un nombre entier en nombre réel et transfère ce réel sur la pile des réels.

Exemple:

```
35 S>F
F. \ affiche 35.000000

3.5e F>S . \ affiche 3
```

Affichage des nombres et chaînes de caractères

Changement de base numérique

FORTH ne traite pas n'importe quels nombres. Ceux que vous avez utilisés en essayant les précédents exemples sont des entiers signés simple précision. Ces nombres peuvent être traités dans n'importe quelle base numérique, toutes les bases numériques situées entre 2 et 36 étant valides :

```
255 HEX . DECIMAL \ affiche FF
```

On peut choisir une base numérique encore plus grande, mais les symboles disponibles sortiront de l'ensemble alpha-numérique [0..9,A..Z] et risquent de devenir incohérents.

La base numérique courante est contrôlée par une variable nommée BASE et dont le contenu peut être modifié. Ainsi, pour passer en binaire, il suffit de stocker la valeur 2 dans BASE. Exemple:

```
2 BASE !
```

et de taper **DECIMAL** pour revenir à la base numérique décimale.

eForth Linux dispose de deux mots pré-définis permettant de sélectionner différentes bases numériques :

- **DECIMAL** pour sélectionner la base numérique décimale. C'est la base numérique prise par défaut au démarrage de eForth Linux;
- HEX pour sélectionner la base numérique hexadécimale ;
- **BINARY** pour sélectionner la base numérique binaire.

Dès sélection d'une de ces bases numériques, les nombres littéraux seront interprétés, affichés ou traités dans cette base. Tout nombre entré précédemment dans une base numérique différente de la base numérique courante est automatiquement converti dans la base numérique actuelle. Exemple :

```
DECIMAL \ base en décimal
255 \ empile 255

HEX \ sélectionne base hexadécimale
1+ \ incrémente 255 devient 256
. \ affiche 100
```

On peut définir sa propre base numérique en définissant le mot approprié ou en stockant cette base dans **BASE**. Exemple :

```
: SEXTAL ( ---) \ sélectionne la base numérique binaire 6 BASE ! ;
DECIMAL 255 SEXTAL . \ affiche 1103
```

Le contenu de BASE peut être empilé comme le contenu de n'importe quelle autre variable :

```
VARIABLE RANGE_BASE \ définition de variable RANGE-BASE

BASE @ RANGE_BASE ! \ stockage contenu BASE dans RANGE-BASE

HEX FF 10 + . \ affiche 10F

RANGE_BASE @ BASE ! \ restaure BASE avec contenu de RANGE-BASE
```

Dans une définition : , le contenu de BASE peut transiter par la pile de retour:

```
: OPERATION ( ---)

BASE @ >R \ stocke BASE sur pile de retour

HEX FF 10 + . \ opération du précédent exemple

R> BASE ! ; \ restaure valeur initiale de BASE
```

ATTENTION: les mots >R et R> ne sont pas exploitables en mode interprété. Vous ne pouvez utiliser ces mots que dans une définition qui sera compilée.

Définition de nouveaux formats d'affichage

Forth dispose de primitives permettant d'adapter l'affichage d'un nombre à un format quelconque. Avec eForth Linux, ces primitives traitent les nombres entiers :

- <# débute une séquence de définition de format ;
- # insère un digit dans une séquence de définition de format ;
- #\$ équivaut à une succession de #;
- HOLD insère un caractère dans une définition de format ;
- #> achève une définition de format et laisse sur la pile l'adresse et la longueur de la chaîne contenant le nombre à afficher.

Ces mots ne sont utilisables qu'au sein d'une définition. Exemple, soit à afficher un nombre exprimant un montant libellé en euros avec la virgule comme séparateur décimal :

```
: .EUROS ( n ---)
<# # # [char] , hold #S #>
type space ." EUR" ;
1245 .euros
```

Exemples d'exécution:

```
35 .EUROS \ affiche 0,35 EUR
3575 .EUROS \ affiche 35,75 EUR
1015 3575 + .EUROS \ affiche 45,90 EUR
```

Dans la définition de .EUROS, le mot <# débute la séquence de définition de format d'affichage. Les deux mots # placent les chiffres des unités et des dizaines dans la chaîne de caractère. Le mot HOLD place le caractère , (virgule) à la suite des deux chiffres de droite, le mot #S complète le format d'affichage avec les chiffres non nuls à la suite de , .

Le mot #> ferme la définition de format et dépose sur la pile l'adresse et la longueur de la chaîne contenant les digits du nombre à afficher. Le mot TYPE affiche cette chaîne de caractères.

En exécution, une séquence de format d'affichage traite exclusivement des nombres entiers 32 bits signés ou non signés. La concaténation des différents éléments de la chaîne se fait de droite à gauche, c'est à dire en commençant par les chiffres les moins significatifs.

Le traitement d'un nombre par une séquence de format d'affichage est exécutée en fonction de la base numérique courante. La base numérique peut être modifiée entre deux digits.

Voici un exemple plus complexe démontrant la compacité du FORTH. Il s'agit d'écrire un programme convertissant un nombre quelconque de secondes au format HH:MM:SS:

```
: :00 ( ---)
DECIMAL # \ insertion digit unité en décimal
6 BASE ! \ sélection base 6
# \ insertion digit dizaine
[char] : HOLD \ insertion caractère :
DECIMAL ; \ retour base décimale

: HMS ( n ---) \ affiche nombre secondes format HH:MM:SS
<# :00 :00 #S #> TYPE SPACE ;
```

Exemples d'exécution:

```
59 HMS \ affiche 0:00:59
60 HMS \ affiche 0:01:00
4500 HMS \ affiche 1:15:00
```

Explication: le système d'affichage des secondes et des minutes est appelé système sexagésimal. Les **unités** sont exprimées dans la base numérique décimale, les **dizaines** sont exprimées dans la base six. Le mot :00 gère la conversion des unités et des dizaines dans ces deux bases pour la mise au format des chiffres correspondants aux secondes et aux minutes. Pour les heures, les chiffres sont tous décimaux.

Autre exemple, soit à définir un programme convertissant un nombre entier simple précision décimal en binaire et l'affichant au format bbbb bbbb bbbb bbbb:

```
R> BASE ! ; \ Initial digital base restoration
```

Exemple d'exécution:

```
DECIMAL 12 AFB \ affiche 0000 0000 0110
HEX 3FC5 AFB \ affiche 0011 1111 1100 0101
```

Encore un exemple, soit à créer un agenda téléphonique où l'on associe à un patronyme un ou plusieurs numéros de téléphone. On définit un mot par patronyme :

```
: .## ( ---)
    # # [char] . HOLD ;
: .TEL ( d ---)
    CR <# .## .## .## .## # # > TYPE CR ;
: DUGENOU ( ---)
    0618051254 .TEL ;
dugenou \ display : 06.18.05.12.54
```

Cet agenda, qui peut être compilé depuis un fichier source, est facilement modifiable, et bien que les noms ne soient pas classés, la recherche y est extrêmement rapide.

Affichage des caractères et chaînes de caractères

L'affichage d'un caractère est réalisé par le mot EMIT:

```
65 EMIT \ affiche A
```

Les caractères affichables sont compris dans l'intervalle 32..255. Les codes compris entre 0 et 31 seront également affichés, sous réserve de certains caractères exécutés comme des codes de contrôle. Voici une définition affichant tout le jeu de caractères de la table ASCII :

```
variable #out
: #out+! ( n -- )
    #out +!
                            \ incrémente #out
: (.) (n -- al)
 DUP ABS <# #S ROT SIGN #>
: .R ( n 1 -- )
 >R (.) R> OVER - SPACES TYPE
: JEU-ASCII ( ---)
    cr 0 #out !
   128 32
   DO
       I 3 .R SPACE
                           \ affiche code du caractère
        4 #out+!
       I EMIT 2 SPACES
                            \ affiche caractère
        3 #out+!
```

```
#out @ 77 =

IF

CR 0 #out!

THEN

LOOP;
```

L'exécution de **JEU-ASCII** affiche les codes ASCII et les caractères dont le code est compris entre 32 et 127. Pour afficher la table équivalente avec les codes ASCII en hexadécimal, taper **HEX JEU-ASCII** :

```
hex jeu-ascii
20 21 ! 22 "
                23 # 24 $ 25 % 26 & 27 '
                                           28 ( 29 )
                                                      2A *
2B + 2C , 2D - 2E . 2F / 30 0 31 1 32 2
                                           33 3 34 4
                                                      35 5
36 6 37 7 38 8 39 9 3A : 3B ; 3C < 3D =
                                                3F ?
                                           3E >
                                                      40 @
41 A 42 B 43 C 44 D 45 E 46 F 47 G 48 H 49 I 4A J
4C L 4D M 4E N 4F O 50 P 51 Q 52 R 53 S 54 T 55 U 56 V
57 W 58 X 59 Y 5A Z 5B [ 5C \ 5D ] 5E ^
                                           5F 60 `
                                                      61 a
62 b 63 c 64 d 65 e 66 f 67 g 68 h 69 i 6A j 6B k
                                                      6C 1
6D m 6E n 6F o 70 p 71 q
                           72 r 73 s
                                      74 t 75 u
                                                76 <del>v</del>
                                                      77 w
                      7C |
                           7D }
                                7E ~
78 x 79 y 7A z 7B {
                                      7F
                                          ok
```

Les chaînes de caractères sont affichées de diverses manières. La première, utilisable en compilation seulement, affiche une chaîne de caractères délimitée par le caractère " (guillemet) :

```
: TITRE ." MENU GENERAL" ;

TITRE \ affiche MENU GENERAL
```

La chaîne est séparée du mot ." par au moins un caractère espace.

Une chaîne de caractères peut aussi être compilée par le mot s'' et délimitée par le caractère " (quillemet) :

```
: LIGNE1 ( --- adr len)
S" E..Enregistrement de données" ;
```

L'exécution de **LIGNE1** dépose sur la pile de données l'adresse et la longueur de la chaîne compilée dans la définition. L'affichage est réalisé par le mot **TYPE** :

```
LIGNE1 TYPE \ affiche E..Enregistrement de données
```

En fin d'affichage d'une chaîne de caractères, le retour à la ligne doit être provoqué s'il est souhaité :

```
CR TITRE CR CR LIGNE1 TYPE CR
\ affiche
\ MENU GENERAL
\
\ E..Enregistrement de données
```

Un ou plusieurs espaces peuvent être ajoutés en début ou fin d'affichage d'une chaîne alphanumérique :

```
SPACE \ affiche un caractère espace

10 SPACES \ affiche 10 caractères espace
```

Variables chaînes de caractères

Les variables alpha-numérique texte n'existent pas nativement dans eForth Linux. Voici le premier essai de définition du mot string :

```
\ define a strvar
: string ( comp: n --- names_strvar | exec: --- addr len )
    create
        dup
        c, \ n is maxlength
        0 c, \ 0 is real length
        allot
    does>
        2 +
        dup 1 - c@
;
```

Une variable chaîne de caractères se définit comme ceci :

```
16 string strState
```

Voici comment est organisé l'espace mémoire réservé pour cette variable texte :

```
string content

current length: 0

max length: 16
```

Code des mots de gestion de variables texte

Voici le code source complet permettant la gestion des variables texte :

```
DEFINED? --str [if] forget --str [then]
create --str

\ compare two strings
: $= ( addr1 len1 addr2 len2 --- fl)
    str=
;

\ define a strvar
: string ( n --- names_strvar )
    create
    dup
    ,
    \ n is maxlength
```

```
\ 0 is real length
        0 ,
       allot
   does>
       cell+ cell+
       dup cell - @
\ get maxlength of a string
: maxlen$ ( strvar --- strvar maxlen )
   over cell - cell - @
\ store str into strvar
: $! ( str strvar --- )
   maxlen$
                            \ get maxlength of strvar
   nip rot min
                           \ keep min length
                         \ store real length
   2dup swap cell - !
                           \ copy string
   cmove
   ;
\ Example:
\ : s1
\ s" this is constant string" ;
\ 200 string test
\ s1 test $!
\ set length of a string to zero
: 0$! ( addr len -- )
   drop 0 swap cell - !
\ extract n chars right from string
: right$ ( str1 n --- str2 )
   0 \text{ max over min } > r + r@ - r>
\ extract n chars left frop string
: left$ ( str1 n --- str2 )
   0 max min
\ extract n chars from pos in string
: mid$ ( str1 pos len --- str2 )
   >r over swap - right$ r> left$
   ;
\ append char c to string
: c+$! ( c str1 -- )
```

La création d'une chaîne de caractères alphanumérique est très simple :

```
64 string myNewString
```

Ici, nous créons une variable alphanumérique myNewString pouvant contenir jusqu'à 64 caractères.

Pour afficher le contenu d'une variable alphanumérique, il suffit ensuite d'utiliser type. Exemple :

```
s" This is my first example.." myNewString $!
myNewString type \ display: This is my first example..
```

Si on tente d'enregistrer une chaîne de caractères plus longue que la taille maximale de notre variable alphanumérique, la chaîne sera tronquée :

```
s" This is a very long string, with more than 64 characters. It can't store
complete"
myNewString $!
myNewString type
  \ affiche: This is a very long string, with more than 64 characters. It
can
```

Ajout de caractère à une variable alphanumérique

Certains périphériques, le transmetteur LoRa par exemple, demandent à traiter des lignes de commandes contenant les caractères non alphanumériques Le mot c+\$! permet cette insertion de code :

```
32 string AT_BAND
s" AT+BAND=868500000" AT_BAND $! \ set frequency at 865.5 Mhz
$0a AT_BAND c+$!
$0d AT_BAND c+$! \ add CR LF code at end of command
```

Le dump mémoire du contenu de notre variable alphanumérique AT_BAND confirme la présence des deux caractères de contrôle en fin de chaine :

```
--> AT_BAND dump
--addr--- 00 01 02 03 04 05 06 07 08 09 0A 0B 0C 0D 0E 0F -----chars----
3FFF-8620 8C 84 FF 3F 20 00 00 00 13 00 00 00 41 54 2B 42 ...? .....AT+B
3FFF-8630 41 4E 44 3D 38 36 38 35 30 30 30 30 0A 0D BD AND=868500000...
ok
```

Voici une manière astucieuse de créer une variable alphanumérique permettant de transmettre un retour chariot, un **CR+LF** compatible avec les fins de commandes pour le transmetteur LoRa:

Les mots de création de mots

FORTH est plus qu'un langage de programmation. C'est un méta-langage. Un méta-langage est un langage utilisé pour décrire, spécifier ou manipuler d'autres langages.

Avec ESP32forth, on peut définir la syntaxe et la sémantique de mots de programmation au-delà du cadre formel des définitions de base.

On a déjà vu les mots définis par **constant**, **variable**, **value**. Ces mots servent à gérer des données numériques.

Dans le chapitre Structures de données pour ESP32forth, on a également utilisé le mot **create**. Ce mot crée un en-tête permettant d'accéder à une zone de données mis en mémoire. Exemple :

```
create temperatures
34 , 37 , 42 , 36 , 25 , 12 ,
```

Ici, chaque valeur est stockée dans la zone des paramètres du mot temperatures avec le mot , .

Avec eForth Linux, on va voir comment personnaliser l'exécution des mots définis par

Utilisation de does>

Il y a une combinaison de mots-clés "CREATE" et "DOES>", qui est souvent utilisée ensemble pour créer des mots (mots de vocabulaire) personnalisés avec des comportements spécifiques.

Voici comment cela fonctionne en Forth:

- CREATE: ce mot-clé est utilisé pour créer un nouvel espace de données dans le dictionnaire ESP32Forth. Il prend en charge un argument, qui est le nom que vous donnez à votre nouveau mot;
- **DOES>**: ce mot-clé est utilisé pour définir le comportement du mot que vous venez de créer avec **CREATE**. Il est suivi d'un bloc de code qui spécifie ce que le mot devrait faire lorsqu'il est rencontré pendant l'exécution du programme.

Ensemble, cela ressemble à quelque chose comme ceci :

Lorsque le mot mon-nouveau-mot est rencontré dans le programme FORTH, le code spécifié dans la partie does> ... ; sera exécuté.

Ici, on définit le mot de définition defREG: qui a exactement la même action que value. Mais pourquoi créer un mot qui recrée l'action d'un mot qui existe déjà ?

```
$00 value DB2INSTANCE
```

ou

```
$00 defREG: DB2INSTANCE
```

sont semblables. Cependant, en créant nos registres avec defREG: on a les avantages suivants :

- un code source eForth Linux plus lisible. On détecte facilement toutes les constantes nommant un registre ESP32;
- on se laisse la possibilité de modifier la partie does> de defREG: sans avoir ensuite à réécrire les lignes de code qui n'utiliseraient pas defREG:

Voici un cas classique, le traitement d'un tableau de données :

L'exécution de temperatures doit être précédé de la position de la valeur à extraire dans ce tableau. Ici nous récupérons seulement l'adresse contenant la valeur à extraire.

Exemple de gestion de couleur

Dans ce premier exemple, on définit le mot **color**: qui va récupérer la couleur à sélectionner et la stocker dans une variable :

```
0 value currentCOLOR
```

L'exécution du mot setBLACK ou setWHITE simplifie considérablement le code eForth Linux. Sans ce mécanisme, il aurait fallu répéter régulièrement une de ces lignes :

```
$00 currentCOLOR !
```

Ou

```
$00 variable BLACK
BLACK currentCOLOR !
```

Exemple, écrire en pinyin

Le pinyin est couramment utilisé dans le monde entier pour enseigner la prononciation du chinois mandarin, et il est également utilisé dans divers contextes officiels en Chine, comme les panneaux de signalisation, les dictionnaires et les manuels d'apprentissage. Il facilite l'apprentissage du chinois pour les personnes dont la langue maternelle utilise l'alphabet latin.

Pour écrire en chinois sur un clavier QWERTY, les Chinois utilisent généralement un système appelé "pinyin input" ou "saisie pinyin". Pinyin est un système de romanisation du chinois mandarin, qui utilise l'alphabet latin pour représenter les sons du mandarin.

Sur un clavier QWERTY, les utilisateurs tapent les sons du mandarin en utilisant la romanisation pinyin. Par exemple, si quelqu'un veut écrire le caractère "你" ("nǐ" signifiant "tu" ou "toi" en français), il peut taper "ni".

Dans ce code très simplifié, on peut programmer des mots pinyin pour écrire en mandarin. Le code ci-après fonctionne parfaitement dans eForth Linux :

```
Work well in eForth Linux
: chinese:
    create ( c1 c2 c3 -- )
        c, c, c,
    does>
        3 type
;
```

Pour trouver le code UTF8 d'un caractère chinois, copiez le caractère chinois, depuis Google Translate par exemple. Exemple :

```
Good Morning --> 早安 (Zao an)
```

Copiez 早 et allez dans eForth Linux et tapez:

```
key key \ followed by key <enter>
```

collez le caractère 早. Eforth Linux doit afficher les codes suivants:

```
230 151 169
```

Pour chaque caractère chinois, on va exploiter ces trois codes ainsi :

```
169 151 230 chinese: Zao
137 174 229 chinese: An
```

Utilisation:

```
Zao An \ display 早安
```

Avouez quand même que programmer ainsi c'est autre chose que ce qu'on peut faire en langage C. Non ?

Traitement des caractères UTF8

C'est en réalisant quelques tests de saisie de caractères au clavier qu'il est apparu un petit souci. Si on fait ceci :

```
key \ and press a key, push 97 on stack
```

Jusque là, tout est normal. Mais sur le clavier, on dispose aussi, en France sur clavier AZERTY, de caractères accentués et certains caractères comme €. Réessayons key en tentant de récupérer le code de ce caractère :

```
key
€
ok
226 --> ��
ERROR: �� NOT FOUND!
```

Le premier code récupéré a la valeur 226. mais il y a deux autres codes qui perturbent l'interpréteur FORTH. Voyons cette solution :

```
key key key €
ok
226 130 172 →
```

Ah... ?!?!? Trois codes ?

Le codage UTF8

Reprenons les trois codes 226 130 172 en hexadécimal : E2 82 AC. Si on fait ceci :

```
$e2 emit
```

Ca affiche • ok. Mmmm.... Vérifions dans un boucle qui est dans l'intervalle 32-255 :

```
: dispChars ( -- )
    256 32 do
    i emit
    loop
;
```

L'exécution de dispChars affiche ceci :

eForth Linux a quelques soucis pour afficher des caractères ayant un code ASCII supérieur à 127. Si on refait ce test avec eForth Windows, le même mot dispChars affiche ceci :

Pour les caractères dont le code ASCII est dans l'intervalle [32..127], les caractères sont identiques. Pour les caractères dont le code ASCII est supérieur à 127 (7F en hexadécimal), eForth Linux ne sait pas afficher de caractères valides.

Pour afficher le caractère € sous eForth Linux, nous avons des solutions simples :

```
: .eur ( -- )
." €" ;
```

ou

```
: .€ ( -- )
." €" ;
```

ou

```
: .eur ( -- )
226 emit 130 emit 172 emit ;
```

Mais nous ne réglons pas le problème des caractères dont le code ASCII est supérieur à 127. Pour résoudre ce problème, il faut s'intéresser au codage UTF8, celui utilisé par eForth Linux.

En codage UTF8, les caractères ASCII sont codés sur 7 bits :

• **Obbb-bbb** codage sur 1 octet

Pour tous les autres caractères, le codage est sur 2 ou 3 ou 4 octets :

• **11**0*b***-***bbbb* 10*bb***-***bbbb* codage sur 2 octets

```
• 1110-bbbb 10bb-bbbb 10bb-bbbb codage sur 3 octets
```

```
• 1111-0bbb 10bb-bbbb 10bb-bbbb 10bb-bbbb codage sur 4 octets
```

Pour tous les codes supérieurs à \$7F, les premiers bits de poids fort déterminent le nombre d'octets codant un caractère UTF8. Revenons à notre caractère €. Le premier code qui remonte en exécutant key est \$E2. En binaire : 11100010. Ici on a trois bits à 1. Ceci signifie que le caractère € est codé sur 3 octets.

Effectuons un test avec le caractère UTF8 ட. Une exécution de **key** fait remonter le code 240, en binaire : **1111**0000. On a 4 bits à 1. Le caractère ட est codé sur quatre octets.

Récupérer le code de caractères UTF8 entrés au clavier

L'idée est de détecter le nombre d'exécution de **key** à exécuter en fonction du caractère saisi au clavier :

- on exécute un premier key
- si le code est supérieur à 127, on fait glisser ce code de 1 bit vers la gauche, puis on teste le bit b7. Si ce bit est à 1 on re-exécute **key**.

Voici le code capable de saisir n'importe quel caractère UTF8 :

```
0 value keyUTF8
: toKeyUTF8 ( c -- )
   keyUTF8 8 lshift or to keyUTF8
;
```

Le mot **toKeyUTF8** reçoit un code clavier sur 8 bits et le concatène au contenu de la valeur **keyUTF8**. L'idée est de récupérer le codage UTF8 en une seule valeur numérique finale.

Le mot **getkeys** traite le code remonté par la première exécution de **key**. Il exécute un glissement d'un octet vers la gauche et teste le bit b7 (séquence **1 lshift dup \$80** and). Si ce bit est à 1, le mot se ré-exécute (séquence **if recurse**).

La récursivité permet de contrôler le nombre d'itérations de **getKeys** sans faire appel à des boucles et tests complexes. La récursivité s'interrompt dès qu'un bit b7 est à 0. La sortie de récursivité s'effectue après **then**. Le mot **getKeys** exécutera la séquence **key toKeyUTF8** autant de fois qu'il y a d'appels récursifs.

Le mot **ukey** peut maintenant se substituer au mot **key** pour récupérer le code UTF8 de n'importe quel caractère du jeu de caractères UTF8 :

```
hex ukey . \ paste € and <enter>, display : E282AC
```

Ce que confirme la documentation UTF8 en ligne.

Affichage de caractères UTF8 depuis leur code

Si on regarde la définition du mot emit, on trouve ceci :

```
: emit
    >R RP@ 1 type rdrop
;
```

La séquence de code RP@ 1 type limite strictement l'affichage d'un caractère code sur un seul octet. Cette séquence hex E282AC emit ne fonctionnera pas. De même :

Le souci vient de l'ordre des octets d'une valeur numérique. Un dump mémoire de la pile donne ceci :

```
--addr--- 00 01 02 03 04 05 06 07 08 09 0A 0B 0C 0D 0E 0F -----chars----- 0802-00FA 00 00 00 00 00 AC 82 E2 00 00 00 00 08 01 ......
```

Il faut donc *retourner* les octets comme une chaussette :

```
\ reverse integer bytes, example:
\ hex 1a2b3C --> 3c2b1a
: reverse-bytes ( n0 -- )
    0 { result }
    3 for
        result 100 * to result
        100 u/mod swap +to result
    next
    drop
    result
;
```

on peut maintenant réécrire notre mot uemit :

```
\ emit UTF8 encoded character
: uemit ( n -- )
    reverse-bytes
    >r rp@ 4 type
    rdrop
;
```

L'exécution de hex E282AC uemit affiche : €.

En conclusion, avec **ukey** et **uemit**, on dispose maintenant de mots permettant de traiter des caractères non-ASCII. Ainsi, avec un clavier grec :

Index lexical

	FCOS46	
BASE48	forget30	struct42
BINARY48	fsqrt45	structures41
c!33	fvariable45	to37
c@33	HEX48	value34
cat19	HOLD49	variable33
cell40	include18	variables locales36
constant34	liste des fichiers18	;30
create57	lshift63	:30
DECIMAL48	mémoire33	."52
DOES>57	mv18	.s28
drop32	pile de retour32	{36
dump28	pinyin59	}36
dup32	recurse62	<i>@</i> 33
effacer fichier18	rm18	#49
EMIT51	S"52	#>49
f44	S>F46	#S49
F**45	see28	+to37
F>S46	set-precision44	<#49
FATAN245	SF!45	
fconstant45	SF@45	